
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.2.4.19.2 which is part of the LTE/SAE test suite in the ‘iwd-TTCN3-B2015-03_D15wk50’ ATS delivery.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
8_2_4_19_2
Test Group:
LTE_IWD_15wk50
ATS Version:
iwd-TTCN3-B2015-03_D15wk50
System Simulator used:
Anite Conformance Toolset solution
UE’s used:
Samsung Zen
Verification Status:
PASS


4. Corrections required for test case 8.2.4.19.2
Change 1 – f_EUTRA_InitialiseWithSameCellTiming
	Function name
	f_EUTRA_InitialiseWithSameCellTiming

	Reason for change
	For TDD-LTE cells the Random Access Response (RAR) Timing Advance (TA) is calculated based on the Tcell value as specified in clause 7.4.3.1. 

According to TS 36.213 clause 4.2.3, RAR TA allowed values shall be in the range 0..1282.

In the case of FDD – TDD CA with PCell as FDD, the Tcell value of the TDD SCell is set same as Tcell value of FDD PCell. For some FDD cells, the Tcell value is such that the calculation defined in clause 7.4.3.1 results in a RAR TA outside the 0..1282 range, hence invalid. This is in reference to CR R5-153143.
This prose CR implementation was missing in f_EUTRA_InitialiseWithSameCellTiming function



	Summary of change
	The TDD RAR TA calculation is updated in f_EUTRA_InitialiseWithSameCellTiming function to ensure that in case of FDD – TDD CA, it remains within the allowed range. 



	TTCN module
	Common\EUTRA\EUTRA_CellInfo.ttcn

	MCC160 Comment
	Accepted


Before change

  
  function f_EUTRA_InitialiseWithSameCellTiming(EUTRA_CellIdList_Type p_EUTRA_CellList) runs on EUTRA_PTC

  { //@sic R5s140757 sic@

    var integer i;

    var integer v_CellListLength := lengthof (p_EUTRA_CellList);

    var CellTimingInfo_Type v_PcellTiming := f_EUTRA_CellInfo_GetCellTimingInfo( p_EUTRA_CellList[0] ); // In CA the first cell is the Pcell

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    // Initialise SFN and Tcell of all cells to the values used in the first Cell in the list

    for (i:=1; i<v_CellListLength; i:=i+1 ) {

      f_EUTRA_CellInfo_Set_SfnOffset(p_EUTRA_CellList[i], v_PcellTiming.SfnOffset);

      f_EUTRA_CellInfo_Set_Tcell(p_EUTRA_CellList[i], v_PcellTiming.Tcell);

      //Set RAR_TA for TDD cells

      v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(p_EUTRA_CellList[i]);

      if (v_EUTRA_FDD_TDD_Mode == TDD) {

       f_EUTRA_CellInfo_SetRAR_TA (p_EUTRA_CellList[i],(v_PcellTiming.Tcell mod 30720)/16);      
      }

    }

  }
After change
       function f_EUTRA_InitialiseWithSameCellTiming(EUTRA_CellIdList_Type p_EUTRA_CellList) runs on EUTRA_PTC

  { //@sic R5s140757 sic@

    var integer i;

    var integer v_CellListLength := lengthof (p_EUTRA_CellList);

    var CellTimingInfo_Type v_PcellTiming := f_EUTRA_CellInfo_GetCellTimingInfo( p_EUTRA_CellList[0] ); // In CA the first cell is the Pcell

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    // Initialise SFN and Tcell of all cells to the values used in the first Cell in the list

    for (i:=1; i<v_CellListLength; i:=i+1 ) {

      f_EUTRA_CellInfo_Set_SfnOffset(p_EUTRA_CellList[i], v_PcellTiming.SfnOffset);

      f_EUTRA_CellInfo_Set_Tcell(p_EUTRA_CellList[i], v_PcellTiming.Tcell);

      //Set RAR_TA for TDD cells

      v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(p_EUTRA_CellList[i]);

      if (v_EUTRA_FDD_TDD_Mode == TDD) {

       f_EUTRA_CellInfo_SetRAR_TA (p_EUTRA_CellList[i],((v_PcellTiming.Tcell mod 30720)/16)mod 1282);
      }

    }

  }
MCC160 additional change
CA test cases executed with FDD/TDD band combinations shall be run with UE of Rel-12 or above.

Added the additional release check in f_EUTRAN_CheckReleaseIndicator 
  //----------------------------------------------------------------------------

  /*

   * @desc      To check the EUTRA release reported by UE against the tc release

   * @param     p_ReleaseIndicator_UE

   * @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_R10_R11, LTE_IRAT, POS)

   */

  function f_EUTRAN_CheckReleaseIndicator(AccessStratumRelease p_ReleaseIndicator_UE)

  { // @sic R5-123750 sic@

    var FreqBandIndicator v_PrimaryFrequencyBand, v_SecondaryFrequencyBand; // @sic R5s160134 sic@
    select (f_GetTestcaseAttrib_Eutra_Release (testcasename())) {

      case ("") { // rel8

      }

      case ("rel_9") { // >= rel9

        if (not match(p_ReleaseIndicator_UE, complement(rel8))) {

          f_SetVerdictInconc (__FILE__, __LINE__, "Wrong Access Stratum release indicator");

        }

      }

      case ("rel_10") { // >= rel10

        if (not match(p_ReleaseIndicator_UE, complement(rel8, rel9))) {

          f_SetVerdictInconc (__FILE__, __LINE__, "Wrong Access Stratum release indicator");

        }

      }

      case ("rel_11") { // >= rel11

        if (not match(p_ReleaseIndicator_UE, complement(rel8, rel9, rel10))) { //@sic R5-142980 sic@

          f_SetVerdictInconc (__FILE__, __LINE__, "Wrong Access Stratum release indicator");

        }

      }

      case ("rel_12") { // >= rel12 //@sic R5-142980: added rel-12 tc sic@

        if (not match(p_ReleaseIndicator_UE, complement(rel8, rel9, rel10, rel11))) {

          f_SetVerdictInconc (__FILE__, __LINE__, "Wrong Access Stratum release indicator");

        }

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    //@sic R5s160134 additional change sic@

    // CA test cases executed with FDD/TDD band combinations shall be run with UE of Rel-12 or above

    if (f_GetTestcaseAttrib_CA_Configured(testcasename())) {

        v_PrimaryFrequencyBand := fl_GetCA_PrimaryBand (px_EUTRA_CA_BandCombination);

        v_SecondaryFrequencyBand := fl_GetCA_SecondaryBand (px_EUTRA_CA_BandCombination);

        if ((f_EUTRA_FrequencyBandIsFDD(v_PrimaryFrequencyBand) != f_EUTRA_FrequencyBandIsFDD(v_SecondaryFrequencyBand)) and 

            match(p_ReleaseIndicator_UE, complement(rel8, rel9, rel10, rel11) )){

                 f_SetVerdictInconc (__FILE__, __LINE__, "Wrong Access Stratum release indicator for an FDD/TDD CA band combination");

        } 

    }

  }
New function in CommonDefs

  // ---------------------------------------------------------------------------------

  /*

   * @desc      Return TRUE if the band is an FDD band or flase otherwise

   * @param     p_FrequencyBand

   * @return    Boolean

   */

   function f_EUTRA_FrequencyBandIsFDD(integer p_FrequencyBand) return boolean

  { //@sic R5s160134 sic@

    return (p_FrequencyBand < 33);

  }

Use the new function in EUTRA_CapabilityFunctions

  function f_Encode_RRCConnectionReconfiguration_HOToEUTRA(CoOrd_EUTRASysInfo_Type p_EUTRASysInfo,

                                                           integer p_N := 1,

                                                           B3_Type p_NasKsi := '000'B) return octetstring

  {

    var bitstring v_Bitstring;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(f_ConvertIntegerToDL_Bandwidth(p_EUTRASysInfo.DL_Bandwidth),

                                                                                                   f_ConvertIntegerToUL_Bandwidth(p_EUTRASysInfo.HO.UL_Bandwidth));

    var template (value) DRB_ToAddModList v_DRB_ToAddModList := { cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1) }

    if (p_N == 2) { //  @sic R5-152145 sic@

      v_DRB_ToAddModList := { cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1), cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB2) }

    }

    if (f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand))   {            // Bands below 33 are reserved for FDD @sic R5s160134 sic@

      v_Bitstring := encvalue(cs_RRCConnectionReconfiguration_HandoverToEUTRA (p_EUTRASysInfo.Arfcn,

                                                                               f_ConvertIntegerToDL_Bandwidth (p_EUTRASysInfo.DL_Bandwidth),

                                                                               f_ConvertIntegerToUL_Bandwidth(p_EUTRASysInfo.HO.UL_Bandwidth),

                                                                               p_EUTRASysInfo.HO.PhysCellId,

                                                                               tsc_RRC_TI_Def,

                                                                               cs_PRACH_Config_Def(cs_508_PRACH_ConfigSIB_FDD ( v_ChannelBandwidthDependency.Prach_FrequencyOffset,

                                                                                                                                p_EUTRASysInfo.HO.RootSequenceIndex )),

                                                                               v_ChannelBandwidthDependency.Pusch_HoppingOffset,

                                                                               cs_508_SoundingRS_UL_ConfigCommon_FDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration),

                                                                               omit,

                                                                               cs_CQI_ReportingPeriodic_FDD,

                                                                               cs_508_PUCCH_ConfigDedicated_Default_FDD,   // @sic R5-123630 sic@

                                                                               v_ChannelBandwidthDependency.sr_PUCCH_ResourceIndex, // @sic R5-123630 sic@

                                                                               30,   // @sic R5-123630 sic@

                                                                               v_ChannelBandwidthDependency.N_RB_CQI, // @sic R5-123712 sic@

                                                                               v_DRB_ToAddModList, // @sic R5-152145 sic@

                                                                               p_NasKsi)); // @sic R5s150704 sic@

    } else {

      v_Bitstring := encvalue(cs_RRCConnectionReconfiguration_HandoverToEUTRA (p_EUTRASysInfo.Arfcn,

                                                                               f_ConvertIntegerToDL_Bandwidth(p_EUTRASysInfo.DL_Bandwidth),

                                                                               omit,

                                                                               p_EUTRASysInfo.HO.PhysCellId,

                                                                               tsc_RRC_TI_Def,

                                                                               cs_PRACH_Config_Def(cs_508_PRACH_ConfigSIB_TDD ( v_ChannelBandwidthDependency.Prach_FrequencyOffset,

                                                                                                                                p_EUTRASysInfo.HO.RootSequenceIndex )),

                                                                               v_ChannelBandwidthDependency.Pusch_HoppingOffset,

                                                                               cs_508_SoundingRS_UL_ConfigCommon_TDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration),

                                                                               cs_TDD_ConfigDef(px_eTDDsubframeConfig),

                                                                               cds_CQI_ReportingPeriodic_TDD,

                                                                               cds_508_PUCCH_ConfigDedicated_Default_TDD,   // @sic R5-123630 sic@

                                                                               v_ChannelBandwidthDependency.sr_PUCCH_ResourceIndex, // @sic R5-123630 sic@

                                                                               27,    // @sic R5-123630 sic@

                                                                               v_ChannelBandwidthDependency.N_RB_CQI, // @sic R5-123712 sic@

                                                                               v_DRB_ToAddModList, // @sic R5-152145 sic@

                                                                               p_NasKsi)); // @sic R5s150704 sic@

    }

    return bit2oct(f_OctetAlignedBitString(v_Bitstring));  // NOTE: bit2oct would add additional bits at the beginning (MSB) whereas here it has be added at the end (LSB)

  }

  //----------------------------------------------------------------------------

  /*

   * @desc      To covert and return the E-UTRA FDD_TDD mode from PIXIT to be used in UTRAN modules

   * @return    EUTRA_FDD_TDD_Mode_Type

   * @status    APPROVED (LTE, LTE_IRAT)

   */

  function  f_EUTRA_ConvertEutraFDDTDD_FromPIXIT() return EUTRA_FDD_TDD_Mode_Type

  {

    if (f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand)) {              // Bands below 33 are reserved for FDD @sic R5s160134 sic@

      return FDD;

    } else {                                 // Bands above 33 are reserved for TDD

      return TDD;

    }

  }

Use the new function in EUTRA_CapabilityFunctions_CA

  function f_Encode_RRCConnectionReconfiguration_HOToEUTRA_CA(CoOrd_EUTRASysInfo_Type p_EUTRASysInfo,

                                                              CoOrd_EUTRASysInfo_Type p_EUTRASysInfo_Scell) return octetstring

  {

    var bitstring v_Bitstring;

    var boolean  v_ISTDD := true; // If EUTRA is FDD or TDD

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(f_ConvertIntegerToDL_Bandwidth(p_EUTRASysInfo.DL_Bandwidth),

                                                                                                   f_ConvertIntegerToUL_Bandwidth(p_EUTRASysInfo.HO.UL_Bandwidth));

    var boolean v_UL_CA := pc_UL_CA;

    var template (omit) CommonSCell_UL_Configuration_Type v_CommonSCell_UL_Configuration := omit;

    var template (omit)  DedicatedSCell_UL_Configuration_Type v_DedicatedSCell_UL_Configuration := omit;

    var template (value) SCellToAddMod_r10 v_SCellToAddMod;

    var SCell_DL_Bandwidth_Type v_SCell_DL_Bandwidth ;

    var template(omit) TDD_Config v_TDD_Config := omit;  // omitted in FDD

    var template (value) PUCCH_ConfigDedicated_v1020 v_PUCCH_ConfigDedicated;

    var template (omit) AdditionalSpectrumEmissionCA_r10_Type v_AdditionalSpectrumEmissionCA_r10 := omit;

    select (p_EUTRASysInfo_Scell.DL_Bandwidth) {

      case (6)   { v_SCell_DL_Bandwidth := n6;  }

      case (15)  { v_SCell_DL_Bandwidth := n15; }

      case (25)  { v_SCell_DL_Bandwidth := n25; }

      case (50)  { v_SCell_DL_Bandwidth := n50; }

      case (75)  { v_SCell_DL_Bandwidth := n75; }

      case (100) { v_SCell_DL_Bandwidth := n100;}

      case else  { FatalError (__FILE__, __LINE__, "invalid BandWidth"); }

    }

    if (f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand)) { //@sic R5s160153 sic@

      v_ISTDD := false; // FDD

    }

    // Scell Info Initialisation

    if (v_ISTDD) {  // initialise tdd-Configuration  to be used in

      v_TDD_Config := cs_TDD_ConfigDef(px_eTDDsubframeConfig);    // Value omitted in FDD

      v_PUCCH_ConfigDedicated  := cs_508_PUCCH_ConfigDedicated_v1020_2TX(cs_N1PUCCH_AN_CS_r10_FDD_1TX, cs_N1PUCCH_AN_CS_r10_FDD_2TX); //@sic R5-133275 sic@

    }

    else { //FDD

      v_PUCCH_ConfigDedicated  := cs_508_PUCCH_ConfigDedicated_v1020_1TX(cs_N1PUCCH_AN_CS_r10_FDD_1TX);

      // Not Addressed for 2TX

    }

    if (v_UL_CA) {

      if (p_EUTRASysInfo_Scell.CA_Tested_Integer == 2) { // inter band, 36.508 Table 4.8.2.1.6-1

        v_AdditionalSpectrumEmissionCA_r10 := cs_AdditionalSpectrumEmissionCA_r10_Def;

      }

      if (v_ISTDD) {

        v_DedicatedSCell_UL_Configuration := cs_RadioResourceConfigDedicatedSCell_UL_Configuration_DEFAULT();//@sic R5s150070 sic@

        v_CommonSCell_UL_Configuration := cs_RadioResourceConfigCommonSCell_UL_Configuration_DEFAULT(v_ChannelBandwidthDependency.Pusch_HoppingOffset,

                                                                                                     cs_508_SoundingRS_UL_ConfigCommon_FDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration)); //

      }

      else {

        v_DedicatedSCell_UL_Configuration := cs_RadioResourceConfigDedicatedSCell_UL_Configuration_DEFAULT();//@sic R5s150070 sic@

        v_CommonSCell_UL_Configuration := cs_RadioResourceConfigCommonSCell_UL_Configuration_DEFAULT(v_ChannelBandwidthDependency.Pusch_HoppingOffset,

                                                                                                     cs_508_SoundingRS_UL_ConfigCommon_TDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration));

      }

    }

    v_SCellToAddMod := cs_SCellToAddMod_r10(1,

                                            p_EUTRASysInfo_Scell.HO.PhysCellId,

                                            p_EUTRASysInfo_Scell.Arfcn,

                                            cs_508_RadioResourceConfigCommonSCell_r10_DEFAULT(v_SCell_DL_Bandwidth,

                                                                                              tsc_P_b_1Tx,

                                                                                              v_CommonSCell_UL_Configuration,

                                                                                              v_TDD_Config),

                                            cs_RadioResourceConfigDedicatedSCell_r10(cs_PhysicalConfigDedicatedSCell_r10_DEFAULT(v_DedicatedSCell_UL_Configuration)));

    if (f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand)) {   // Bands below 33 are reserved for FDD @sic R5s160134 sic@

      v_Bitstring := encvalue(cs_RRCConnectionReconfiguration_HandoverToEUTRA_CA(p_EUTRASysInfo.Arfcn,

                                                                                 f_ConvertIntegerToDL_Bandwidth (p_EUTRASysInfo.DL_Bandwidth),

                                                                                 f_ConvertIntegerToUL_Bandwidth(p_EUTRASysInfo.HO.UL_Bandwidth),

                                                                                 p_EUTRASysInfo.HO.PhysCellId,

                                                                                 tsc_RRC_TI_Def,

                                                                                 cs_PRACH_Config_Def(cs_508_PRACH_ConfigSIB_FDD(v_ChannelBandwidthDependency.Prach_FrequencyOffset,

                                                                                                                                p_EUTRASysInfo.HO.RootSequenceIndex)),

                                                                                 v_ChannelBandwidthDependency.Pusch_HoppingOffset,

                                                                                 cs_508_SoundingRS_UL_ConfigCommon_FDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration),

                                                                                 omit,

                                                                                 //cs_CQI_ReportingPeriodic_FDD,//@sic R5s130979 sic@

                                                                                 cs_508_PUCCH_ConfigDedicated_Default_FDD,   // @sic R5-123630 sic@

                                                                                 v_ChannelBandwidthDependency.sr_PUCCH_ResourceIndex, // @sic R5-123630 sic@

                                                                                 30,   // @sic R5-123630 sic@

                                                                                 v_ChannelBandwidthDependency.N_RB_CQI,

                                                                                 v_PUCCH_ConfigDedicated,

                                                                                 v_AdditionalSpectrumEmissionCA_r10,

                                                                                 -,

                                                                                 -,

                                                                                 {v_SCellToAddMod})); // @sic R5-123712 sic@

    } else {

      v_Bitstring := encvalue(cs_RRCConnectionReconfiguration_HandoverToEUTRA_CA(p_EUTRASysInfo.Arfcn,

                                                                                 f_ConvertIntegerToDL_Bandwidth(p_EUTRASysInfo.DL_Bandwidth),

                                                                                 omit,

                                                                                 p_EUTRASysInfo.HO.PhysCellId,

                                                                                 tsc_RRC_TI_Def,

                                                                                 cs_PRACH_Config_Def(cs_508_PRACH_ConfigSIB_TDD ( v_ChannelBandwidthDependency.Prach_FrequencyOffset,

                                                                                                                                  p_EUTRASysInfo.HO.RootSequenceIndex )),

                                                                                 v_ChannelBandwidthDependency.Pusch_HoppingOffset,

                                                                                 cs_508_SoundingRS_UL_ConfigCommon_TDD(v_ChannelBandwidthDependency.SrsBandwidthConfiguration),

                                                                                 cs_TDD_ConfigDef(px_eTDDsubframeConfig),

                                                                                 //cds_CQI_ReportingPeriodic_TDD,//@sic R5s130979 sic@

                                                                                 cds_508_PUCCH_ConfigDedicated_Default_TDD,   // @sic R5-123630 sic@

                                                                                 v_ChannelBandwidthDependency.sr_PUCCH_ResourceIndex, // @sic R5-123630 sic@

                                                                                 27,    // @sic R5-123630 sic@

                                                                                 v_ChannelBandwidthDependency.N_RB_CQI,

                                                                                 v_PUCCH_ConfigDedicated,

                                                                                 v_AdditionalSpectrumEmissionCA_r10,

                                                                                 -,

                                                                                 -,

                                                                                 {v_SCellToAddMod})); // @sic R5-123712 sic@

    }

    return bit2oct(f_OctetAlignedBitString(v_Bitstring));  // NOTE: bit2oct would add additional bits at the beginning (MSB) whereas here it has be added at the end (LSB)

  }

}

Use the new function in LTE_36523_SelectionExpressions

  function f_SelectionExpr(LTE_36523_Condition_Type p_Condition) return boolean 

  { /* Set the applicability conditions according to 36.523-2 Table 4-1a: Applicability of tests Conditions */

    var boolean v_ApplCond;

    select (p_Condition) {

      //R: UEs supporting E-UTRA

      case (R) {

        v_ApplCond := true;

      }

      //C01: UEs supporting E-UTRA and UTRA

      case (C01) {

        v_ApplCond := pc_UTRA;  //@sic R5s140528 sic@

      }

      //C01a: UEs supporting E-UTRA and UTRA FDD

      case(C01a){

         v_ApplCond := pc_FDD; //@sic R5-142484 R5s140528 sic@

      }

      //C02: UEs supporting E-UTRA and combined EPS/IMSI attach (with or without pre-configuration)

      case (C02) {

        v_ApplCond := pc_Combined_Attach;

      }

      //C03: UEs supporting USIM removal without power down

      case (C03) {

        v_ApplCond := pc_USIM_Removal;

      }

      //C04: UEs supporting E-UTRA and EPS attach (with or without pre-configuration)

      case (C04) {

        v_ApplCond := pc_Attach;

      }

      //C05: UEs supporting GERAN

      case (C05) {

        v_ApplCond := pc_GERAN;

      }

      //C06: UEs supporting E-UTRA and HRPD

      case (C06) {

        v_ApplCond := pc_HRPD;

      }

      //C07: UEs supporting E-UTRA and 1xRTT

      case (C07) {

        v_ApplCond := pc_1xRTT;

      }

      //C08: UEs supporting E-UTRA and Feature Group 5

      case (C08) {

        if ( f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand) ) {       //@sic R5-150557 sic@

            v_ApplCond := pc_FeatrGrp_5_F;

        } else {

            v_ApplCond := pc_FeatrGrp_5_T;

        }

      }

      //C10: UEs supporting E-UTRA and Feature Group Indicator 25

      case (C10) {

        if ( f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand) ) {        //@sic R5-150557 sic@

            v_ApplCond := pc_FeatrGrp_25_F;

        } else {

            v_ApplCond := pc_FeatrGrp_25_T;

        }

      }

      //C11: UEs supporting E-UTRA and Feature Group Indicator 16 and Feature Group Indicator 25

      case (C11) {

        if ( f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand) ) {        //@sic R5-150557 sic@

            v_ApplCond := pc_FeatrGrp_16_F and pc_FeatrGrp_25_F;

        } else {

            v_ApplCond := pc_FeatrGrp_16_T and pc_FeatrGrp_25_T;

        }

      }

      //C13: UEs supporting E-UTRA and UTRA and Feature Group Indicator 16 and Feature Group Indicator 22

      case (C13) {

        if ( f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand) ) {        //@sic R5-150557 sic@

            v_ApplCond := pc_UTRA and pc_FeatrGrp_16_F and pc_FeatrGrp_22_F;

        } else {

            v_ApplCond := pc_UTRA and pc_FeatrGrp_16_T and pc_FeatrGrp_22_T;

        }

      }

      //C14: UEs supporting E-UTRA and Feature Group Indicator 5 and Feature Group Indicator 17

      case (C14) {

         if ( f_EUTRA_FrequencyBandIsFDD(px_ePrimaryFrequencyBand) ) {        //@sic R5-150557 sic@

            v_ApplCond := pc_FeatrGrp_5_F and pc_FeatrGrp_17_F;

         } else {

            v_ApplCond := pc_FeatrGrp_5_T and pc_FeatrGrp_17_T;

         }

      }

Etc…

5. Execution Log Files
1) Samsung ZEN
The Samsung ZEN UE passed this test case on Anite Conformance Toolset Solution on CA FDD+TDD band combination CA_1A-41A. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_8_2_4_19_2_Log.html
	
	


